
1. Overview

This document lists all the essencial changes needed to correct problems in the TTCN implementation of test cases 7.1.2.3.10 which is part of the ENDC test suite in the ‘iwd-TTCN3-B2018-12_D19wk12_5G’ ATS delivery. 
With these changes applied the test case can be demonstrated to run with one or more UEs (see section 4). Execution log files are provided as evidence.

3. Verification Test Summary

Test Case:
7.1.2.3.10
Test Group:
\ENDC\7_1_2\
ATS Version:
iwd-TTCN3-B2018-12_D19wk12_5G
System Simulator used:
Anritsu Protocol Conformance Test System ME7834NR
UE used:
Qualcomm SDM855+SDX50
Verification Status:
PASS


4. Corrections required for ENDC TC 7.1.2.3.10 using wk12 (Anritsu)
Change 4.1
	Function name
	f_TC_7_1_2_3_10_NR_TestBody

	Reason for change
	1) As per updated prose, Grants sizes 224(Step 22A), 240(Step 22), 408(Step 18), 432(Step 18A) and 768 ( Step 15) are updated. 
2) Step 2 and Step 4 are lastPDU sent by UE so, polling bit will be sent
3) TX_Next and RX_Next are updated so that SN=0 and SN=1 are sent and received accordingly at step 1 to 4. Similarly step 9, 12, 17 and 19.

4) Delay calculated at step 21 based on the timingInfo Received in Step 20.
5) Added to allow successful RACH

6) PDCP header added inside fl_NR_GenerateRLC_AM_SDUs. Hence, 3 octets are substracted
7) Delay from 17 would be 20ms, but it is STATUS PDU sent by SS at step 17. Therefore 100ms is used from step 16. Similary at step 22.
8) Step 21: Use new Status PDU template to NACK the SDU with SN=3
9) Step 24: 2nd Segment transmitted by UE for the NACKed segment should contain SegmentationInfo ‘11’, so adjust the expected value

10) Step 10 to 11: RLC STATUS PDU should ACK_SN=3;


	Summary of change
	1) Adjust sizes and grants accordingly
2) Expect Poll

3) AM_TX_Next and AM_RX_Next fixes
4) Step 19 and 20: Store the TimingInfo of last received PDU 
5) Add delay
6) Reduce the generated RLC SDU size by 3
7) Step 18 and 22: Increase the waiting time before grant is generated from 20ms to 100ms
8) Step 21: Use new Status PDU template to NACK the SDU with SN=3
9) Step 24: 2nd Segment transmitted by UE for the NACKed segment should contain SegmentationInfo ‘11’, so adjust the expected value
10) Step 10 to 11: RLC STATUS PDU should ACK_SN=3;


	TTCN module
	RLC_TC_Common_NR.ttcn

	MCC160 Comment
	


Before change

  function f_TC_7_1_2_3_10_NR_TestBody(NR_RLC_SS_State_Type p_RLC_Rec,

                                             DRB_Identity p_NR_DRB_Id) runs on NR_BASE_PTC

  { // AM RLC / Re-transmission of RLC PDU with and without re-segmentation

    var integer v_RLC_Len1stSegPart1 := 27;

    var integer v_RLC_Len1stSegPart2 := 22;
    var integer v_RLC_Len1stSeg := v_RLC_Len1stSegPart1 + v_RLC_Len1stSegPart2; // = 49

    var integer v_RLC_Len2ndSeg := 47;
    var integer v_RLC_SDUsize := v_RLC_Len1stSeg + v_RLC_Len2ndSeg; // = 96

    var SubFrameTiming_Type v_TimingDL;

    var SubFrameTiming_Type v_TimingUL;

    var template (value) NR_RLC_Status_NackListSN18Bit_Type v_NR_NackSN18Bit_List;

    var NR_ResourceAllocation_Type v_NR_ResourceAllocation;

    var NR_UplinkBWP_Type v_NR_UplinkBWP := f_NR_CellInfo_GetUplinkBWP(nr_Cell1, tsc_NR_BWP_Id);

    var integer i;

    // generate 4 RLC SDUs = PDCP PDUs for transmission

    fl_NR_GenerateRLC_AM_SDUs(p_RLC_Rec, v_RLC_SDUsize, 4);

    // @siclog "Step 1" siclog@

    // The SS transmits one AMD PDU containing SDU#1 (96 bytes) in its data field.

    f_NR_TxAMD_SDU(p_RLC_Rec, p_NR_DRB_Id);

    // @siclog "Step 2" siclog@

    // The UE transmits one AMD PDU containing SDU#1 in its data field.

    f_NR_RxAMD_SDU (p_RLC_Rec, p_NR_DRB_Id, tsc_RLC_SDU1);

    // @siclog "Step 3" siclog@

    // The SS transmits one AMD PDU containing SDU#2 (96 bytes) in its data field.

    f_NR_TxAMD_SDU(p_RLC_Rec, p_NR_DRB_Id);

    // @siclog "Step 4" siclog@

    // The UE transmits one AMD PDU containing SDU#2 in its data field.

    f_NR_RxAMD_SDU (p_RLC_Rec, p_NR_DRB_Id, tsc_RLC_SDU2);

    // @siclog "Step 5" siclog@

    // The SS transmits a RLC STATUS PDU. ACK_SN=2, NACK_SN=0.

    v_NR_NackSN18Bit_List  := {crs_NR_RLC_NACK_SN18_Simple (0)};

    f_NR_TxSTATUS_PDU(p_RLC_Rec, p_NR_DRB_Id, v_NR_NackSN18Bit_List);

    // @siclog "Step 6" siclog@

    // Check: Does the UE transmit the AMD PDU not yet acknowledged?

    p_RLC_Rec.AM_RX_Next := 0;

    f_NR_RxAMD_SDU (p_RLC_Rec, p_NR_DRB_Id, tsc_RLC_SDU1);

    f_NR_PreliminaryPass(__FILE__, __LINE__, "Step 6");

    // @siclog "Step 7" siclog@

    // The SS transmits a RLC STATUS PDU. ACK_SN=2.

    p_RLC_Rec.AM_RX_Next := 2;

    f_NR_TxSTATUS_PDU(p_RLC_Rec, p_NR_DRB_Id);

    // @siclog "Step 8" siclog@

    // The SS transmits one AMD PDU containing SDU#3 (96 bytes) in its data field.

    f_NR_TxAMD_SDU(p_RLC_Rec, p_NR_DRB_Id);

    // @siclog "Step 9" siclog@

    // The UE transmits an AMD PDU containing SDU#3 in its data field.

    f_NR_RxAMD_SDU (p_RLC_Rec, p_NR_DRB_Id, tsc_RLC_SDU3);

    // EXCEPTION: Steps 10 to 11 shall be repeated until RETX_COUNT=maxRetxThreshold-1.

    for (i := 0; i < (p_RLC_Rec.maxRetxThreshold-1); i:= i + 1) {

      // @siclog "Step 10" siclog@

      // The SS transmits a RLC STATUS PDU. ACK_SN=3 and NACK_SN=2.

      v_NR_NackSN18Bit_List  := {crs_NR_RLC_NACK_SN18_Simple (2)};

      f_NR_TxSTATUS_PDU(p_RLC_Rec, p_NR_DRB_Id, v_NR_NackSN18Bit_List);

      // @siclog "Step 11" siclog@

      // Check: Does the UE retransmit the AMD PDU not yet acknowledged?

      p_RLC_Rec.AM_RX_Next := 2;

      f_NR_RxAMD_SDU (p_RLC_Rec, p_NR_DRB_Id, tsc_RLC_SDU3);

      f_NR_PreliminaryPass(__FILE__, __LINE__, "Step 11");

    }

    // @siclog "Step 12" siclog@

    // The SS transmits a RLC STATUS PDU. ACK_SN=3.

    f_NR_TxSTATUS_PDU(p_RLC_Rec, p_NR_DRB_Id);

    // @siclog "Step 13" siclog@

    // The SS stops the UL grant transmission.

    f_NR_ULGrantConfiguration_Stop(nr_Cell1, -, cs_NR_UplinkTimeAlignment_Keep);

    // @siclog "Step 14" siclog@

    // The SS transmits one AMD PDU containing SDU#4 (96 bytes) in its data field.

    v_TimingDL := f_NR_GetNextSendOccasion_DL(nr_Cell1);

    f_NR_TxAMD_SDU(p_RLC_Rec, p_NR_DRB_Id, tsc_NR_P_NoPoll, cs_TimingInfo_NR(v_TimingDL));

    // @siclog "Step 15" siclog@

    // The SS waits for 60 ms and allocates one UL grant of size 808 bits.

    v_TimingUL := f_NR_GetNextSendOccasion_ULafterDL(nr_Cell1, v_TimingDL, 60);

    v_NR_ResourceAllocation := f_NR_ResourceAllocation_UplinkGrant(808, v_NR_UplinkBWP);
    f_NR_OneULGrantTransmission(nr_Cell1, cs_TimingInfo_NR(v_TimingUL), v_NR_ResourceAllocation);

    // @siclog "Step 16" siclog@

    // The UE transmits an AMD PDU with the same data contents as received in the corresponding part of SDU#4?

    f_NR_RxAMD_SDU (p_RLC_Rec, p_NR_DRB_Id, tsc_RLC_SDU4);

    // @siclog "Step 17" siclog@

    // The SS transmits a STATUS PDU. This PDU nacks the AMD PDU with SN=3. ACK_SN=4 and NACK_SN=3.

    v_NR_NackSN18Bit_List  := {crs_NR_RLC_NACK_SN18_Simple (3)};

    f_NR_TxSTATUS_PDU(p_RLC_Rec, p_NR_DRB_Id, v_NR_NackSN18Bit_List);

    // @siclog "Step 18" siclog@

    // The SS waits for 20 ms and then allocates 2 UL grants of size 432 bits such that there is 20 ms gap between UL grants.

    v_TimingUL := f_NR_GetNextSendOccasion_ULafterDL(nr_Cell1, p_RLC_Rec.TimeStampLastReceivedPDU, 20);

    v_NR_ResourceAllocation := f_NR_ResourceAllocation_UplinkGrant(432, v_NR_UplinkBWP);

    f_NR_ULGrantConfiguration_StartPeriodicCyclicGrant(nr_Cell1, cs_TimingInfo_NR(v_TimingUL), v_NR_ResourceAllocation, 2, 20);
    // @siclog "Step 19" siclog@

    // Check: Does the UE transmit an SDU segment with SI=01 and SOEnd=48 and the same data contents at the received positions as in the original SDU#4?

    p_RLC_Rec.AM_RX_Next := 3;

    f_NR_RxAMD_PDUpartofSDU (p_RLC_Rec, p_NR_DRB_Id, tsc_RLC_SDU4, ?, 0, v_RLC_Len1stSeg, tsc_NR_SI_StartOfSDU);

    f_NR_PreliminaryPass(__FILE__, __LINE__, "Step 19");

    // @siclog "Step 20" siclog@

    // Check: Does the UE transmit an SDU segment with SI=10 and SOStart=49 and the same data contents at the received positions as in the original SDU#4?

    f_NR_RxAMD_PDUpartofSDU (p_RLC_Rec, p_NR_DRB_Id, tsc_RLC_SDU4, ?, v_RLC_Len1stSeg, v_RLC_Len2ndSeg, tsc_NR_SI_EndOfSDU, true);

    f_NR_PreliminaryPass(__FILE__, __LINE__, "Step 20");

    // @siclog "Step 21" siclog@

    // After 100 ms SS transmits a STATUS PDU. This PDU nacks the SDU with SN=3. NACK_SN=3, SOStart=0, SOEnd=48 and ACK_SN=4

    v_TimingDL := f_SubFrameTiming_AddMilliSeconds(p_RLC_Rec.TimeStampLastReceivedPDU, 100);

    v_NR_NackSN18Bit_List := {crs_NR_RLC_NACK_SN18_SOs (3, 0, (v_RLC_Len1stSeg-1))};

    p_RLC_Rec.AM_RX_Next := 4;

    f_NR_TxSTATUS_PDU(p_RLC_Rec, p_NR_DRB_Id, v_NR_NackSN18Bit_List);

    // @siclog "Step 22" siclog@

    // The SS waits for 20 ms and then allocates 2 UL grants (UL grant allocation type 2) of size 256 bits such that there is 20 ms gap between UL grants.

    v_TimingUL := f_NR_GetNextSendOccasion_ULafterDL(nr_Cell1, p_RLC_Rec.TimeStampLastReceivedPDU, 20);

    v_NR_ResourceAllocation := f_NR_ResourceAllocation_UplinkGrant(256, v_NR_UplinkBWP);

    f_NR_ULGrantConfiguration_StartPeriodicCyclicGrant(nr_Cell1, cs_TimingInfo_NR(v_TimingUL), v_NR_ResourceAllocation, 2, 20);
    // @siclog "Step 23" siclog@

    // Check: Does the UE transmit an AMD PDU segment with SI=01 and SOEnd=26 and the same data contents at the received positions as in the original SDU#4?

    p_RLC_Rec.AM_RX_Next := 3;

    f_NR_RxAMD_PDUpartofSDU (p_RLC_Rec, p_NR_DRB_Id, tsc_RLC_SDU4, ?, 0, v_RLC_Len1stSegPart1, tsc_NR_SI_StartOfSDU);

    f_NR_PreliminaryPass(__FILE__, __LINE__, "Step 23");

    // @siclog "Step 24" siclog@

    // Check: Does the UE transmit an AMD PDU segment with SI=10, SOStart=27 and the same data contents at the received positions as in the original SDU#4?

    f_NR_RxAMD_PDUpartofSDU (p_RLC_Rec, p_NR_DRB_Id, tsc_RLC_SDU4, ?, v_RLC_Len1stSegPart1, v_RLC_Len1stSegPart2, tsc_NR_SI_EndOfSDU, true);

    f_NR_PreliminaryPass(__FILE__, __LINE__, "Step 24");

    // @siclog "Step 25" siclog@

    // The SS transmits a STATUS PDU. This PDU acks the AMD PDUs with SN=3. ACK_SN=4.

    p_RLC_Rec.AM_RX_Next := 4;

    f_NR_TxSTATUS_PDU(p_RLC_Rec, p_NR_DRB_Id);

  }
  }

After change
  function f_TC_7_1_2_3_10_NR_TestBody(NR_RLC_SS_State_Type p_RLC_Rec,

                                             DRB_Identity p_NR_DRB_Id) runs on NR_BASE_PTC

  { // AM RLC / Re-transmission of RLC PDU with and without re-segmentation

    var integer v_RLC_Len1stSegPart1 := 23;

    var integer v_RLC_Len1stSegPart2 := 21;
    var integer v_RLC_Len1stSeg := v_RLC_Len1stSegPart1 + v_RLC_Len1stSegPart2; // = 49

    var integer v_RLC_Len2ndSeg := 46;
    var integer v_RLC_SDUsize := v_RLC_Len1stSeg + v_RLC_Len2ndSeg; // = 96

    var SubFrameTiming_Type v_TimingDL;

    var SubFrameTiming_Type v_TimingUL;

    var template (value) NR_RLC_Status_NackListSN18Bit_Type v_NR_NackSN18Bit_List;

    var NR_ResourceAllocation_Type v_NR_ResourceAllocation;

    var NR_UplinkBWP_Type v_NR_UplinkBWP := f_NR_CellInfo_GetUplinkBWP(nr_Cell1, tsc_NR_BWP_Id);

    var integer i;

    f_Delay(1.0);
    // generate 4 RLC SDUs = PDCP PDUs for transmission

    fl_NR_GenerateRLC_AM_SDUs(p_RLC_Rec, (v_RLC_SDUsize-3), 4);

    // @siclog "Step 1" siclog@

    // The SS transmits one AMD PDU containing SDU#1 (96 bytes) in its data field.

    f_NR_TxAMD_SDU(p_RLC_Rec, p_NR_DRB_Id);

    // @siclog "Step 2" siclog@

    // The UE transmits one AMD PDU containing SDU#1 in its data field.

    f_NR_RxAMD_SDU (p_RLC_Rec, p_NR_DRB_Id, tsc_RLC_SDU1, tsc_NR_P_Poll); // ADDED
    p_RLC_Rec.AM_TX_Next := 1;

    p_RLC_Rec.AM_RX_Next := 1;
    // @siclog "Step 3" siclog@

    // The SS transmits one AMD PDU containing SDU#2 (96 bytes) in its data field.

    f_NR_TxAMD_SDU(p_RLC_Rec, p_NR_DRB_Id);

    // @siclog "Step 4" siclog@

    // The UE transmits one AMD PDU containing SDU#2 in its data field.

    f_NR_RxAMD_SDU (p_RLC_Rec, p_NR_DRB_Id, tsc_RLC_SDU2, tsc_NR_P_Poll); // ADDED
    p_RLC_Rec.AM_TX_Next := 2;

    p_RLC_Rec.AM_RX_Next := 2; 

    // @siclog "Step 5" siclog@

    // The SS transmits a RLC STATUS PDU. ACK_SN=2, NACK_SN=0.

    v_NR_NackSN18Bit_List  := {crs_NR_RLC_NACK_SN18_Simple (0)};

    f_NR_TxSTATUS_PDU(p_RLC_Rec, p_NR_DRB_Id, v_NR_NackSN18Bit_List);

    // @siclog "Step 6" siclog@

    // Check: Does the UE transmit the AMD PDU not yet acknowledged?

    p_RLC_Rec.AM_RX_Next := 0;

    f_NR_RxAMD_SDU (p_RLC_Rec, p_NR_DRB_Id, tsc_RLC_SDU1, tsc_NR_P_Poll);

    f_NR_PreliminaryPass(__FILE__, __LINE__, "Step 6");

    // @siclog "Step 7" siclog@

    // The SS transmits a RLC STATUS PDU. ACK_SN=2.

    p_RLC_Rec.AM_RX_Next := 2;

    f_NR_TxSTATUS_PDU(p_RLC_Rec, p_NR_DRB_Id);

    // @siclog "Step 8" siclog@

    // The SS transmits one AMD PDU containing SDU#3 (96 bytes) in its data field.

    f_NR_TxAMD_SDU(p_RLC_Rec, p_NR_DRB_Id);

    // @siclog "Step 9" siclog@

    // The UE transmits an AMD PDU containing SDU#3 in its data field.

    f_NR_RxAMD_SDU (p_RLC_Rec, p_NR_DRB_Id, tsc_RLC_SDU3, tsc_NR_P_Poll); // ADDED
    p_RLC_Rec.AM_TX_Next := 3;

    p_RLC_Rec.AM_RX_Next := 3;
    // EXCEPTION: Steps 10 to 11 shall be repeated until RETX_COUNT=maxRetxThreshold-1.

    for (i := 0; i < (p_RLC_Rec.maxRetxThreshold-1); i:= i + 1) {

      // @siclog "Step 10" siclog@

      // The SS transmits a RLC STATUS PDU. ACK_SN=3 and NACK_SN=2.


p_RLC_Rec.AM_RX_Next := 3;
      v_NR_NackSN18Bit_List  := {crs_NR_RLC_NACK_SN18_Simple (2)};

      f_NR_TxSTATUS_PDU(p_RLC_Rec, p_NR_DRB_Id, v_NR_NackSN18Bit_List);

      // @siclog "Step 11" siclog@

      // Check: Does the UE retransmit the AMD PDU not yet acknowledged?

      p_RLC_Rec.AM_RX_Next := 2;

      f_NR_RxAMD_SDU (p_RLC_Rec, p_NR_DRB_Id, tsc_RLC_SDU3, tsc_NR_P_Poll); // ADDED
      f_NR_PreliminaryPass(__FILE__, __LINE__, "Step 11");

    }

    // @siclog "Step 12" siclog@

    // The SS transmits a RLC STATUS PDU. ACK_SN=3.

    p_RLC_Rec.AM_RX_Next := 3;
    f_NR_TxSTATUS_PDU(p_RLC_Rec, p_NR_DRB_Id);

    // @siclog "Step 13" siclog@

    // The SS stops the UL grant transmission.

    f_NR_ULGrantConfiguration_Stop(nr_Cell1, -, cs_NR_UplinkTimeAlignment_Keep);

    // @siclog "Step 14" siclog@

    // The SS transmits one AMD PDU containing SDU#4 (96 bytes) in its data field.

    v_TimingDL := f_NR_GetNextSendOccasion_DL(nr_Cell1);

    f_NR_TxAMD_SDU(p_RLC_Rec, p_NR_DRB_Id, tsc_NR_P_NoPoll, cs_TimingInfo_NR(v_TimingDL));

    // @siclog "Step 15" siclog@

    // The SS waits for 60 ms and allocates one UL grant of size 808 bits.

    v_TimingUL := f_NR_GetNextSendOccasion_ULafterDL(nr_Cell1, v_TimingDL, 60);

    v_NR_ResourceAllocation := f_NR_ResourceAllocation_UplinkGrant(768, v_NR_UplinkBWP);
    f_NR_OneULGrantTransmission(nr_Cell1, cs_TimingInfo_NR(v_TimingUL), v_NR_ResourceAllocation);

    // @siclog "Step 16" siclog@

    // The UE transmits an AMD PDU with the same data contents as received in the corresponding part of SDU#4?

    f_NR_RxAMD_SDU (p_RLC_Rec, p_NR_DRB_Id, tsc_RLC_SDU4, tsc_NR_P_Poll); // ADDED
    // @siclog "Step 17" siclog@

    // The SS transmits a STATUS PDU. This PDU nacks the AMD PDU with SN=3. ACK_SN=4 and NACK_SN=3.

    v_NR_NackSN18Bit_List  := {crs_NR_RLC_NACK_SN18_Simple (3)};

    p_RLC_Rec.AM_RX_Next := 4;
    f_NR_TxSTATUS_PDU(p_RLC_Rec, p_NR_DRB_Id, v_NR_NackSN18Bit_List);

    // @siclog "Step 18" siclog@

    // The SS waits for 20 ms and then allocates 2 UL grants of size 432 bits such that there is 20 ms gap between UL grants.

    v_TimingUL := f_NR_GetNextSendOccasion_ULafterDL(nr_Cell1, p_RLC_Rec.TimeStampLastReceivedPDU, 100);

    v_NR_ResourceAllocation := f_NR_ResourceAllocation_UplinkGrant(408, v_NR_UplinkBWP);

    f_NR_OneULGrantTransmission(nr_Cell1, cs_TimingInfo_NR(v_TimingUL), v_NR_ResourceAllocation);

    v_TimingUL := f_NR_GetNextSendOccasion_ULafterDL(nr_Cell1, p_RLC_Rec.TimeStampLastReceivedPDU, 120);

    v_NR_ResourceAllocation := f_NR_ResourceAllocation_UplinkGrant(432, v_NR_UplinkBWP);  

    f_NR_OneULGrantTransmission(nr_Cell1, cs_TimingInfo_NR(v_TimingUL), v_NR_ResourceAllocation);
    // @siclog "Step 19" siclog@

    // Check: Does the UE transmit an SDU segment with SI=01 and SOEnd=48 and the same data contents at the received positions as in the original SDU#4?

    p_RLC_Rec.AM_RX_Next := 4; 

    p_RLC_Rec.TimeStampLastReceivedPDU := f_NR_RxAMD_PDUpartofSDU (p_RLC_Rec, p_NR_DRB_Id, tsc_RLC_SDU4, ?, 0, v_RLC_Len1stSeg, tsc_NR_SI_StartOfSDU);

    f_NR_PreliminaryPass(__FILE__, __LINE__, "Step 19");

    // @siclog "Step 20" siclog@

    // Check: Does the UE transmit an SDU segment with SI=10 and SOStart=49 and the same data contents at the received positions as in the original SDU#4?

    p_RLC_Rec.TimeStampLastReceivedPDU := f_NR_RxAMD_PDUpartofSDU (p_RLC_Rec, p_NR_DRB_Id, tsc_RLC_SDU4, ?, v_RLC_Len1stSeg, v_RLC_Len2ndSeg, tsc_NR_SI_EndOfSDU, true);

    f_NR_PreliminaryPass(__FILE__, __LINE__, "Step 20");

    // @siclog "Step 21" siclog@

    // After 100 ms SS transmits a STATUS PDU. This PDU nacks the SDU with SN=3. NACK_SN=3, SOStart=0, SOEnd=48 and ACK_SN=4

    v_TimingDL := f_SubFrameTiming_AddMilliSeconds(p_RLC_Rec.TimeStampLastReceivedPDU, 100);

    v_NR_NackSN18Bit_List := {crs_NR_RLC_NACK_SN18_SOs_712310 (3, 0, (v_RLC_Len1stSeg-1))};

    p_RLC_Rec.AM_RX_Next := 4;

    f_NR_TxSTATUS_PDU(p_RLC_Rec, p_NR_DRB_Id, v_NR_NackSN18Bit_List);

    // @siclog "Step 22" siclog@

    // The SS waits for 20 ms and then allocates 2 UL grants (UL grant allocation type 2) of size 256 bits such that there is 20 ms gap between UL grants.

    v_TimingUL := f_NR_GetNextSendOccasion_ULafterDL(nr_Cell1, p_RLC_Rec.TimeStampLastReceivedPDU, 100);
    v_NR_ResourceAllocation := f_NR_ResourceAllocation_UplinkGrant(240, v_NR_UplinkBWP);

    f_NR_OneULGrantTransmission(nr_Cell1, cs_TimingInfo_NR(v_TimingUL), v_NR_ResourceAllocation);

    v_TimingUL := f_NR_GetNextSendOccasion_ULafterDL(nr_Cell1, p_RLC_Rec.TimeStampLastReceivedPDU, 120);

    v_NR_ResourceAllocation := f_NR_ResourceAllocation_UplinkGrant(224, v_NR_UplinkBWP);  

    f_NR_OneULGrantTransmission(nr_Cell1, cs_TimingInfo_NR(v_TimingUL), v_NR_ResourceAllocation);
    // @siclog "Step 23" siclog@

    // Check: Does the UE transmit an AMD PDU segment with SI=01 and SOEnd=26 and the same data contents at the received positions as in the original SDU#4?

    p_RLC_Rec.AM_RX_Next := 4;
    f_NR_RxAMD_PDUpartofSDU (p_RLC_Rec, p_NR_DRB_Id, tsc_RLC_SDU4, ?, 0, v_RLC_Len1stSegPart1, tsc_NR_SI_StartOfSDU);

    f_NR_PreliminaryPass(__FILE__, __LINE__, "Step 23");

    // @siclog "Step 24" siclog@

    // Check: Does the UE transmit an AMD PDU segment with SI=10, SOStart=27 and the same data contents at the received positions as in the original SDU#4?

    f_NR_RxAMD_PDUpartofSDU (p_RLC_Rec, p_NR_DRB_Id, tsc_RLC_SDU4, ?, v_RLC_Len1stSegPart1, v_RLC_Len1stSegPart2, tsc_NR_SI_SegmentOfSDU, true);

    f_NR_PreliminaryPass(__FILE__, __LINE__, "Step 24");

    // @siclog "Step 25" siclog@

    // The SS transmits a STATUS PDU. This PDU acks the AMD PDUs with SN=3. ACK_SN=4.

    p_RLC_Rec.AM_RX_Next := 4;

    f_NR_TxSTATUS_PDU(p_RLC_Rec, p_NR_DRB_Id);

  }
Change 4.2
	Function name
	f_NR_RxAMD_PDUpartofSDU

	Reason for change
	As per 5) in 4.1, TimingInfo is returned to get timingInfo of the previously RxPDU

	Summary of change
	Return SubFrameTiming_Type for Change 4.1.4


	TTCN module
	RLC_TC_Common_NR.ttcn

	MCC160 Comment
	


Before change

 function f_NR_RxAMD_PDUpartofSDU(NR_RLC_SS_State_Type p_RLC_Rec,

                                   DRB_Identity p_NR_DRB_Id,

                                   integer p_SDUn,

                                   template (present) B1_Type p_Poll,

                                   integer p_Offset,

                                   integer p_Len,

                                   NR_RLC_SegmentationInfo_Type p_SegmentationInfo,

                                   boolean p_OffsetIndication := false)

    runs on NR_BASE_PTC
  {

    var NR_RLC_DataField_Type v_RLC_Data1;

    var template (omit) NR_RLC_SegmentOffset_Type v_SegmentOffset := omit;

    var template (present) NR_RLC_AMD_PDU_Type v_NR_RLC_AMD_PDU;

    var template (present) NR_RLC_PDUList_Type v_NR_RLC_PDUList;

    v_RLC_Data1 := substr (p_RLC_Rec.NR_RLC_DataList[p_SDUn], p_Offset, p_Len);

    if (p_OffsetIndication == true) {

      v_SegmentOffset := int2bit(p_Offset, 16);

    }

    v_NR_RLC_AMD_PDU := f_NR_RLC_Get_AMD_PDU_RX(p_RLC_Rec, p_Poll, p_SegmentationInfo, v_SegmentOffset, v_RLC_Data1);

    v_NR_RLC_PDUList := { cr_NR_RLC_PDU_AMD(v_NR_RLC_AMD_PDU) };

    DRB.receive (car_NR_DRB_COMMON_IND_RLC_PDUList(nr_Cell1,

                                                   p_NR_DRB_Id,

                                                   crs_RlcBearerRouting_None,

                                                   -,

                                                   v_NR_RLC_PDUList));

  }
After change

 function f_NR_RxAMD_PDUpartofSDU(NR_RLC_SS_State_Type p_RLC_Rec,

                                   DRB_Identity p_NR_DRB_Id,

                                   integer p_SDUn,

                                   template (present) B1_Type p_Poll,

                                   integer p_Offset,

                                   integer p_Len,

                                   NR_RLC_SegmentationInfo_Type p_SegmentationInfo,

                                   boolean p_OffsetIndication := false)

   runs on NR_BASE_PTC return SubFrameTiming_Type
  {

    var NR_RLC_DataField_Type v_RLC_Data1;

    var template (omit) NR_RLC_SegmentOffset_Type v_SegmentOffset := omit;

    var template (present) NR_RLC_AMD_PDU_Type v_NR_RLC_AMD_PDU;

    var template (present) NR_RLC_PDUList_Type v_NR_RLC_PDUList;

    var NR_DRB_COMMON_IND v_NR_DRB_COMMON_IND;

    v_RLC_Data1 := substr (p_RLC_Rec.NR_RLC_DataList[p_SDUn], p_Offset, p_Len);

    if (p_OffsetIndication == true) {

      v_SegmentOffset := int2bit(p_Offset, 16);

    }

    v_NR_RLC_AMD_PDU := f_NR_RLC_Get_AMD_PDU_RX(p_RLC_Rec, p_Poll, p_SegmentationInfo, v_SegmentOffset, v_RLC_Data1);

    v_NR_RLC_PDUList := { cr_NR_RLC_PDU_AMD(v_NR_RLC_AMD_PDU) };

    DRB.receive (car_NR_DRB_COMMON_IND_RLC_PDUList(nr_Cell1,

                                                   p_NR_DRB_Id,

                                                   crs_RlcBearerRouting_None,

                                                   -,

                                                   v_NR_RLC_PDUList)) -> value v_NR_DRB_COMMON_IND;
    return v_NR_DRB_COMMON_IND.Common.TimingInfo.SubFrame;
  }
Change 4.3
	Function name
	template (value) NR_RLC_Status_NackSN18Bit_Type crs_NR_RLC_NACK_SN18_SOs_712310

	Reason for change
	Define template of RLC Status PDU specific for 7.1.2.3.10

	Summary of change
	Define template of RLC Status PDU specific for 7.1.2.3.10

	TTCN module
	NR_RLC_Common.ttcn

	MCC160 Comment
	


New template
 template (value) NR_RLC_Status_NackSN18Bit_Type crs_NR_RLC_NACK_SN18_SOs_712310(integer p_NackSN,

                                                                           integer p_SOstart,

                                                                           integer p_SOstop) :=

  { /* serves to transmit or receive  */

    SequenceNumberNACK := int2bit(p_NackSN, 18),       /* 18 bits SN */

    E1 := '0'B,                         /* 1 bit E1 field */

    E2 := '1'B,                         /* 1 bit E2 field */ 

    E3 := '0'B,                         /* 1 bit E3 field */

    Reserved := '000'B,                 /* 1 bits reserved */

    SOstart := int2bit(p_SOstart, 16),

    SOstop := int2bit(p_SOstop, 16),

    NACKrange := omit

  };
Change 4.4
	Function name
	f_TC_7_1_2_3_10_ENDC_NR

	Reason for change
	Change pollByte value to default value as per 38.508-1

	Summary of change
	Change pollByte value to default value as per 38.508-1

	TTCN module
	RLC_ENDC_NR.ttcn

	MCC160 Comment
	


Before Change

  function f_TC_7_1_2_3_10_ENDC_NR() runs on ENDC_NR_PTC

  { // AM RLC / Re-transmission of RLC PDU with and without re-segmentation

    var NR_RLC_SS_State_Type v_RLC_Rec;

    var DL_t_Reassembly_Type v_T_Reassembly;

    var template (omit)  NR_RadioBearerList_Type v_SS_Drb_ConfigList;

    var template (value) NR_RLC_BearerToAddModList_Type v_RLC_BearerConfigList;

    var template (value) DRB_ToAddModList  v_DRB_ToAddModList;

    f_ENDC_NR_Init();

    //Create and configure NR cell1

    f_NR_CellConfig_Def(nr_Cell1);

    f_NR_ENDC_PreambleOnEUTRA(nr_Cell1, MCG_Only);

    v_SS_Drb_ConfigList[0] := cs_NR_SS_RadioBearer_AM_Test(tsc_NR_DRB2); // SNsize 18 by default

    v_DRB_ToAddModList[0] := cs_38508_DRB_ToAddMod(tsc_NR_DRB2, cs_38508_PDCP_Config);

    v_T_Reassembly := f_NR_DL_AM_T_Reassembly(f_NR_CellInfo_GetIsFR1(nr_Cell1));

    v_RLC_BearerConfigList := f_NR_ReconfigureRLC_AM(tsc_NR_DRB2,

                                                     -,

                                                     ms150, // t_PollRetransmit

                                                     -,

                                                     -,
                                                     -,

                                                     -,

                                                     v_T_Reassembly);

    // Get NR capabilities sent during EUTRA preamble

    f_NR_ENDC_ConfigAM_UM(nr_Cell1, 1, 0, v_SS_Drb_ConfigList, v_RLC_BearerConfigList, -, v_DRB_ToAddModList);

    // Initialization of RLC variables

    f_NR_InitRLC_Record(v_RLC_Rec);

    v_RLC_Rec.AM_SN_Size := 18;

    v_RLC_Rec.t_PollRetransmit := 0.15;

    if (not f_NR_CellInfo_GetIsFR1(nr_Cell1)) {v_RLC_Rec.t_Reassembly := 0.03;};

    f_NR_TestBody_Set(true);

    f_TC_7_1_2_3_10_NR_TestBody (v_RLC_Rec, tsc_NR_DRB2);

    f_EUTRA_NR_SendCoOrd (EUTRA, cms_EUTRA_NR_Trigger);

    f_NR_TestBody_Set(false);

    //Wait from coordination AFTER the connection is released in E-UTRA

    f_EUTRA_NR_WaitForCoOrd_Trigger(EUTRA, "Postamble");

    //Postamble

    f_NR_ReleaseAllCells();

  }

After Change

  function f_TC_7_1_2_3_10_ENDC_NR() runs on ENDC_NR_PTC

  { // AM RLC / Re-transmission of RLC PDU with and without re-segmentation

    var NR_RLC_SS_State_Type v_RLC_Rec;

    var DL_t_Reassembly_Type v_T_Reassembly;

    var template (omit)  NR_RadioBearerList_Type v_SS_Drb_ConfigList;

    var template (value) NR_RLC_BearerToAddModList_Type v_RLC_BearerConfigList;

    var template (value) DRB_ToAddModList  v_DRB_ToAddModList;

    f_ENDC_NR_Init();

    //Create and configure NR cell1

    f_NR_CellConfig_Def(nr_Cell1);

    f_NR_ENDC_PreambleOnEUTRA(nr_Cell1, MCG_Only);

    v_SS_Drb_ConfigList[0] := cs_NR_SS_RadioBearer_AM_Test(tsc_NR_DRB2); // SNsize 18 by default

    v_DRB_ToAddModList[0] := cs_38508_DRB_ToAddMod(tsc_NR_DRB2, cs_38508_PDCP_Config);

    v_T_Reassembly := f_NR_DL_AM_T_Reassembly(f_NR_CellInfo_GetIsFR1(nr_Cell1));

    v_RLC_BearerConfigList := f_NR_ReconfigureRLC_AM(tsc_NR_DRB2,

                                                     -,

                                                     ms150, // t_PollRetransmit

                                                     -,

                                                     kB750,
                                                     -,

                                                     -,

                                                     v_T_Reassembly);

    // Get NR capabilities sent during EUTRA preamble

    f_NR_ENDC_ConfigAM_UM(nr_Cell1, 1, 0, v_SS_Drb_ConfigList, v_RLC_BearerConfigList, -, v_DRB_ToAddModList);

    // Initialization of RLC variables

    f_NR_InitRLC_Record(v_RLC_Rec);

    v_RLC_Rec.AM_SN_Size := 18;

    v_RLC_Rec.t_PollRetransmit := 0.15;

    if (not f_NR_CellInfo_GetIsFR1(nr_Cell1)) {v_RLC_Rec.t_Reassembly := 0.03;};

    f_NR_TestBody_Set(true);

    f_TC_7_1_2_3_10_NR_TestBody (v_RLC_Rec, tsc_NR_DRB2);

    f_EUTRA_NR_SendCoOrd (EUTRA, cms_EUTRA_NR_Trigger);

    f_NR_TestBody_Set(false);

    //Wait from coordination AFTER the connection is released in E-UTRA

    f_EUTRA_NR_WaitForCoOrd_Trigger(EUTRA, "Postamble");

    //Postamble

    f_NR_ReleaseAllCells();

  }

5. Execution Log Files

5.1 Qualcomm SDM855+SDX50
The Qualcomm SDM855+SDX50 UE passed this test case on Anritsu Protocol Conformance Test System ME7834NR on EN-DC band combination DC_5A_n261A.The documentation below is enclosed as evidence of the successful test case run [1]:

· Test Case Execution log file

(Note: PICS/PIXIT settings are captured at the beginning of the TLI log)


In the log file (in .txt / -html format) the complete test case execution can be seen. All TLI events are presented and message contents are fully decoded and can be verified. Preliminary verdicts and the final test case verdict can be seen in the log file.

